
* * * First Change * * * *
[bookmark: _Toc145654605][bookmark: _Toc145654417][bookmark: _Toc145654388]5.2.3.2.6	3gpp-Sbi-Binding
This header contains a comma-delimited list of Binding Indications from an HTTP server for storage and subsequent use by an HTTP client (see clause 6.12).
The encoding of the header follows the ABNF as defined in IETF RFC 9110 [11].

Sbi-Binding-Header = "3gpp-Sbi-Binding:" OWS 1#( "bl=" blvalue 1*( ";" OWS parameter ) [ ";" OWS recoverytime ] [ ";" OWS notif-receiver ] [ ";" OWS "group=" groupvalue ] [ 1*( ";" OWS groupparameter ) ] [ ";" OWS "no-redundancy=" no-red-value ] [";" OWS "callback-uri-prefix”" ]) OWS

blvalue = "nf-instance" / "nf-set" / "nfservice-instance" / "nfservice-set"

parameter =  parametername "=" token

parametername = "nfinst" / "nfset" / "nfservinst" / "nfserviceset" / "servname" / "scope" / "backupamfinst" / "backupnf"

recoverytime = "recoverytime=" OWS DQUOTE date-time DQUOTE

notif-receiver = "nr=" URI	; URI production rule from IETF RFC 3986, Appendix A

groupvalue = "true" / "false"

groupparameter = groupparametername "=" token

groupparametername = "oldgroupid" / "groupid" / "uribase" / "oldnfinst / "oldservset" / "oldservinst" / "guami"

no-red-value = "true"


The following parameters are defined:
-	scope: indicates the applicability of a Binding Indication in a service request other than a notification request, or in a notification or callback response. This may take one of the following values:
-	"other-service": the binding information applies to other service(s) that the NF Service Consumer may later on provide as an NF Service Producer (see clause 6.12.3);
-	"subscription-events": the binding information applies to subscription change event notifications (see clause 6.12.4);
-	"callback": the binding information applies to notification or callback requests (see clauses 6.12.4 and 6.12.5).
[bookmark: _PERM_MCCTEMPBM_CRPT57490016___3]The absence of this parameter in a Binding Indication in a service request other than a notification request, or in a notification or callback response, shall be interpreted as "callback".

Two scope parameters may be present in a Binding Indication if the binding information applies to notification/callback requests and to other services.
-	servname (service name): indicates the name of a service, as defined in 3GPP TS 29.510 [8], or a custom service, i.e.:
-	the name of the service that handles a notification or a callback request, when present in a Binding Indication for a subscription or a callback, i.e. with a scope parameter absent or set to "callback"; or
-	the name of the other service(s) for which the binding applies, when present in a Binding Indication in a service request for the other services the NF Service Consumer can provide later on as an NF Service Producer, i.e. with the scope parameter set to "other-service". More than one servname parameter may be present to represent multiple such services. The absence of this parameter in a Binding Indication with the scope parameter set to "other-service" shall be interpreted as binding information that applies to all the services that the NF Service Consumer may provide later as an NF Service Producer.
-	recoverytime: indicates the recovery timestamp of the entity corresponding to the highest resiliency level supported for the resource, that is, the higher level binding entity indicated in the Binding Indication. See Table 6.3.1.0-1 of 3GPP TS 23.501 [3] and clause 6.1 of 3GPP TS 23.527 [38]. The date-time type is specified in IETF RFC 5322 [37] and clause 5.6.7 of IETF RFC 9110 [11].
-	nr: indicates the URI of the notification endpoint when this binding information is applicable; it applies to callback requests (see clause 6.12.4); if the notification URI does not contain a correlationID in the path (i.e. it is a common notification URI for multiple subscriptions), the correlationID shall be added as a fragment component of the URI (i.e. following the "#" character) at the end of the URI.
-	for the definition and encoding of the blvalue, nfinst, backupamfinst, nfset, nfservinst and nfserviceset see clause 5.2.3.2.5.
-	backupnf: indicates the backup NF service instance identifier and/or the backup NF identifier as defined in clause 5.2.2.2.2 or in 3GPP TS 29.510 [8], which shall be used when preferred binding entity is not reachable if supported.
-	group: it is a boolean indicating if the binding indication is for a group of resource/session contexts.
-	groupid (group id): indicates the group identifier allocated by the NF (service) instance, one ore more resource/session contexts are sharing the same groupid. The groupid is optional and it may be allocated when the resource/session context is created and then be updated afterwards. The groupid is global unique and it may be encoded using the same mechnaism for the NfInstanceId as specificed in 3GPP TS 29.571 [13].
-	oldgroupid (old group id): indicates the group identifier allocated by the NF (service) instance previously and to be replaced by the groupid, hence it shall only be present when to update a Binding Indication for multiple contexts.  When the if the oldgroupid is present, the groupid shall also be present to indicate the new groupid allocated.
-	uribase: identify the apiroot and path segments part in the resource URI or notification/callback URI which is common to multiple contexts. This parameter may only be present when to update a Binding Indication for multiple contexts and when the "group" is set to "true". When included, it indicates that all resources or notification contexts with this uribase will use the updated Binding Indication subsequently. More than one uribase may be present.
-	oldnfinst: indicates the NF Instance ID of the NF instance where the group of resource/session contexts are currently served (i.e. the Binding Indication allocated previously for the group of resource/session contexts includes the information of the NF instance), as defined in clause 5.2.2.2.2 in 3GPP TS 29.510 [8]. When included, it indicates that all the resource/session contexts served by this NF instance will use the updated Binding Indication subsequently.
-	oldservset: indicates the NF Service Set ID of the NF Service Set where the group of resource/session contexts are currently served (i.e. the Binding Indication allocated previously for the group of resource/session contexts includes the information of the NF Service Set), as defined in clause 5.2.2.2.2 in 3GPP TS 29.510 [8]. When included, it indicates that all the resource/session contexts served by this NF Service Set will use the updated Binding Indication subsequently.
-	oldservinst: indicates the NF Service Instance ID of the NF service instance where the group of resource/session contexts are currently served (i.e. the Binding Indication allocated previously for the group of resource/session contexts includes the information of the NF service instance), as defined in clause 5.2.2.2.2 in 3GPP TS 29.510 [8]. When included, it indicates that all the resource/session contexts served by this NF service instance will use the updated Binding Indication subsequently.
-	guami (GUAMI): indicates the GUAMI of the AMF currently serving UE contexts, as defined in clause 5.3.4.1 of 3GPP TS 29.571 [13]. When included, it indicates that all the UE contexts associated with the GUMAI will use the updated Binding Indication subsequently.
-	no-redundancy: it is a boolean set to true indicating that the resource is exclusively bound to the NF service instance identified in the binding indication. It may be present in a binding with any scope, i.e. "other-service", "subscription-events" or "callback", or with no scope parameter. When this parameter is present, the blvalue shall be set to "nfservice-instance", the nfservinst parameter shall be present and either the nfserviceset parameter or the nfinst parameter shall be present. The nfserviceset or nfinst parameter included in the binding indication shall only be used to identify the NF service instance and shall not be considered as a binding entity for reselection. The no-redundancy parameter shall only be signaled if the receiver of this information is known to support this parameter (see clause 6.12.1). Subsequently, when sending further requests targeting a resource with no-redundancy, the HTTP client shall not include any routing binding indication in the request message (to prevent the SCP from performing any reselection).
-	callback-uri-prefix: The NF service consumer may include this parameter when providing a Callback URI when the Callback URI contains a prefix. When present, the "callback-uri-prefix" shall be a path-absolute as specified IETF RFC 3986 [14] (i.e. the first path segment(s) after the authority) which is part of the Callback URI provided by a NF service consumer in the corresponding service request message sent to a NF service producer. The authority and "callback-uri-prefix" in the Callback URI shall uniquely identify a consumer service instance. This parameter is relevant when the "scope" parameter is either "subscription-events" or  "callback". See clauses 6.12.4 and 6.12.5 for the usage of this parameter. The ABNF is defined in clause 5.2.3.3.7.
EXAMPLES 1 to 5:	Same as EXAMPLES 1 to 5 defined in clause 5.2.3.2.5, with the header name "3gpp-Sbi-Binding" instead of "3gpp-Sbi-Routing-Binding".
EXAMPLE 6:	Subscription request from one NF on behalf of another NF, with 2 binding indications:

3gpp-Sbi-Binding: bl=nf-set; nfset=set1.udmset.5gc.mnc012.mcc345; servname=nudm-ee;scope=subscription-events
3gpp-Sbi-Binding: bl=nf-set; nfset=set1.nefset.5gc.mnc012.mcc345; servname=nnef-event-exposure
EXAMPLE 7:	Service request with 2 binding indications, for callback requests and for other services the NF Service Consumer may provide later as an NF Service Producer:

3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804518-4191-46b3-955c-ac631f953ed8; nfset=set1.smfset.5gc.mnc012.mcc345; servname=nsmf-pdusession
3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804518-4191-46b3-955c-ac631f953ed8; nfset=set1.smfset.5gc.mnc012.mcc345; scope=other-service; servname=nsmf-event-exposure
EXAMPLE 8:	Service request with one binding indication applying to notification/callback requests and to any other services the NF Service Consumer may provide later as an NF Service Producer:

3gpp-Sbi-Binding: bl=nf-set; nfset=set1-region48.amfset.5gc.mnc012.mcc345; scope=callback; scope=other-service
EXAMPLE 9:	Service request with one binding indication applying to notification/callback requests together with a recovery time stamp associated with the NF Set indicated in the binding indication and with the binding level set to "nfset":
3gpp-Sbi-Binding: bl=nf-set; nfset=set1-region48.amfset.5gc.mnc012.mcc345; scope=callback; recoverytime= "Tue, 04 Feb 2020 08:49:37 GMT"
EXAMPLE 10:	Service response with one binding indication applying to the session context with a recovery time stamp associated with the NF Set indicated in "nfset" in the binding indication and with the binding level set to "nfinstance":

3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804518-4191-46b3-955c-ac631f953ed8; nfset=set1.smfset.5gc.mnc012.mcc345; recoverytime= "Tue, 04 Feb 2020 08:49:37 GMT"
EXAMPLE 11:	Service response with one binding indication applying to the session context with a recovery time stamp associated with the NF Instance included the binding indication and with the binding level set to nfserviceinstance:

3gpp-Sbi-Binding: bl=nfservice-instance; nfservinst=xyz; nfinst=54804518-4191-46b3-955c-ac631f953ed8; recoverytime= "Tue, 04 Feb 2020 08:49:37 GMT"
EXAMPLE 12:	Service response with one binding indication applying to the resource context pertaining to a group identified by "54804518-4191-46b3-955c-ac631f953ed1" together with a backup nf:

3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804518-4191-46b3-955c-ac631f953ed0; nfset=set1.smfset.5gc.mnc012.mcc345; backupnf=54804519-4191-46b3-955c-ac631f953ed2; groupid=54804518-4191-46b3-955c-ac631f953ed1
EXAMPLE 13:	A notification request message with one binding indication applying to the resource contexts with the oldgroup identifier "54804518-4191-46b3-955c-ac631f953ed1", where the preferred binding entity is changed to "nfinst=54804519-4191-46b3-955c-ac631f953ed0" together with a new group identifier "54804519-4191-46b3-955c-ac631f953ed3" allocated.

3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804519-4191-46b3-955c-ac631f953ed0; nfset=set1.smfset.5gc.mnc012.mcc345; group=true; oldgroupid=54804518-4191-46b3-955c-ac631f953ed1; groupid=54804519-4191-46b3-955c-ac631f953ed3
EXAMPLE 14:	A notification request message with one binding indication applying to the resource contexts identified by an uribase, where the preferred binding entity is changed to "nfinst=54804519-4191-46b3-955c-ac631f953ed0":

3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804519-4191-46b3-955c-ac631f953ed0; nfset=set1.smfset.5gc.mnc012.mcc345; group=true; uribase=http%3A%2F%2F10.10.10.10%2Fstringxyz
EXAMPLE 15:	A notification request message with one binding indication applying to the resource contexts served by the NF instance identified by "64804518-4191-46b3-955c-ac631f953ed8" where the preferred binding entity is changed to "nfinst=74804519-4191-46b3-955c-ac631f953ed0".

3gpp-Sbi-Binding: bl=nf-instance; nfinst=74804519-4191-46b3-955c-ac631f953ed0; nfset=set1.smfset.5gc.mnc012.mcc345; group=true; oldnfinst=64804518-4191-46b3-955c-ac631f953ed8
EXAMPLE 16:	Service request message with an updated binding indication applying to the UE contexts for GUAMI" <mnc(012)><mcc(345)><AmfId("abcd12")> where the backupamfinst is changed.

3gpp-Sbi-Binding: bl=nf-instance; nfinst=54804518-4191-46b3-955c-ac631f953ed7; backupamfinst=54804520-4191-46b3-955c-ac631f953ed8; scope=other-service; group=true; guami=%7B%22plmnId%22%3A%7B%22mnc%22%3A%22012%22%2C%22mcc%22%3A%22345%22%7D%2C%22amfId%22%3A%22abcd12%22%7D
EXAMPLE 17:	Service response with a binding indication applying to the resource context which is exclusively bound to a specific NF service instance.
	3gpp-Sbi-Binding: bl=nfservice-instance; nfservinst=xyz; nfinst=54804518-4191-46b3-955c-ac631f953ed8; no-redundancy=true
EXAMPLE 18:	Subscription request with a Callback URI containing a prefix "/abc":

3gpp-Sbi-Binding: bl=nf-set; nfset=set1.nefset.5gc.mnc012.mcc345; servname=nnef-event-exposure; callback-uri-prefix="/abc"
NOTE:	Examples 6 and 7 are formatted as two distinct headers (which improves the readability), but they can also be formatted as a single header with two Binding Indication values separated by a comma.

* * * Next Change * * * *
5.2.3.3.12	3gpp-Sbi-Request-Info
The header contains a list of additional information related to a HTTP request which may be included by a NF or a SCP, to indicate e.g.:
-	whether the HTTP request message is involving a reselection of an alternative NF;
-	whether the HTTP request message is a retransmission of the message, i.e. the request message has been sent but being rejected with a temporary failure or timeout;
When the header is included by a NF acting as a HTTP client, an idempotency-key may be included for a non-idempotent request to enable the receiver to detect possible duplicated request messages as specified in clause 5.2.8.
The receiving NF may use the header, e.g. to determine whether to accept the request.
The encoding of the header follows the ABNF as defined in IETF RFC 9110 [11].

Sbi-Request-Info-Header = "3gpp-Sbi-Request-Info:" OWS req-param *( ";" OWS req-param ) OWS

req-param = req-param-name "=" OWS req-param-value

req-param-name = "retrans" / "redirect" / "reason" / "idempotency-key" / "receivedrejectioncause" / “"callback-uri-prefix”" / token

req-param-value = token



The following parameters are defined:
-	"reason": indicates the reason for which the NF resends or redirects the HTTP request message. This may take one of the following values:
-	"unreachable": indicates that the HTTP request is redirected to an alternative NF due to the request URI (e.g. the resource URI or Notification/callback URI) is not reachable;
-	"overloaded": indicates that the HTTP request is redirected to an alternative NF as result of overload control enforcement, by doing redirection towards an alternative NF (see clause 6.4.3.5.1);
-	"3xx-redirect": indicates that the HTTP request is redirected to an alternative NF as result of receiving a 3xx status code.
-	"temporary-rejection-cause": indicates the HTTP request is retransmitted towards the same or alternative NF due to a temporary rejection.
-	"receivedrejectioncause": indicates a temporary rejection application cause received from the NF or SCP (for last attempt) as defined in clause 5.2.7.2, when the "retrans" parameter is set to "true" and the reason is set to "temporary-rejection-cause". The cause data type is specified in clause 5.2.4.1 of 3GPP TS 29.571 [13].
-	"retrans": it is a boolean and shall be set to "true" to indicate that the request message has been retransmitted e.g. when the request didn't get any response or get a temporary failure cause, otherwise the "retrans" shall not be present.
-	"redirect": it is a boolean and shall be set to "true" to indicate that the request message has been redirected to an alternative NF.
-	"idempotency-key": it is a string and may be encoded using Universally Unique Identifier (UUID), as described in IETF RFC 4122 [47], to uniquely identify a request message (to be received) in the target NF. See clause 5.2.8.
-	"callback-uri-prefix": path-absolute as specified IETF RFC 3986 [14] (i.e. the first path segment(s) after the authority) which is part of the Callback URI.
EXAMPLE 1:	For a request retransmitted to an alternative NF due to the rejection by the original target NF with a temporary rejection cause:

3gpp-Sbi-Request-Info: retrans=true; redirect=true; reason=temporary-rejection-cause; receivedrejectioncause=INSUFFICIENT_RESOURCES
EXAMPLE 2:	For a request sent towards an alternative NF due to the original target NF not reachable:

3gpp-Sbi-Request-Info: redirect=true; reason=unreachable
EXAMPLE 3:	For a non-idempotent request:
	3gpp-Sbi-Request-Info: idempotency-key=54804518-4191-46b3-955c-ac631f953ed8

* * * Next Change * * * *
D.2	ABNF definitions (Filename: "TS29500_CustomHeaders.abnf")

(… text not shown for clarity …)


; ----------------------------------------
;   RFC 9110
; ----------------------------------------

OWS            = *( SP / HTAB )

RWS            = 1*( SP / HTAB )

tchar          = "!" / "#" / "$" / "%" / "&" / "'" / "*" / "+" / "-"
               / "." / "^" / "_" / "`" / "|" / "~" / DIGIT / ALPHA

token          = 1*tchar

BWS            = OWS

quoted-string  = DQUOTE *( qdtext / quoted-pair ) DQUOTE

qdtext         = HTAB / SP / %x21 / %x23-5B / %x5D-7E / obs-text

obs-text       = %x80-FF


date1          = day-rfc9110 SP month-rfc9110 SP year-rfc9110 ; e.g., 02 Jun 1982

day-rfc9110    = 2DIGIT

month-rfc9110  = %x4A.61.6E ; "Jan", case-sensitive
               / %x46.65.62 ; "Feb", case-sensitive
               / %x4D.61.72 ; "Mar", case-sensitive
               / %x41.70.72 ; "Apr", case-sensitive
               / %x4D.61.79 ; "May", case-sensitive
               / %x4A.75.6E ; "Jun", case-sensitive
               / %x4A.75.6C ; "Jul", case-sensitive
               / %x41.75.67 ; "Aug", case-sensitive
               / %x53.65.70 ; "Sep", case-sensitive
               / %x4F.63.74 ; "Oct", case-sensitive
               / %x4E.6F.76 ; "Nov", case-sensitive
               / %x44.65.63 ; "Dec", case-sensitive

year-rfc9110   = 4DIGIT

codings        = content-coding / "identity" / "*"

content-coding = token

weight         = OWS ";" OWS "q=" qvalue

qvalue         = ( "0" [ "." *3DIGIT ] ) / ( "1" [ "." *3"0" ] )


credentials    = auth-scheme [ 1*SP ( token68
               / [ ( "," / auth-param ) *( OWS "," [ OWS auth-param ] ) ] ) ]

auth-scheme    = token

auth-param     = token BWS "=" BWS ( token / quoted-string )

token68        = 1*( ALPHA / DIGIT / "-" / "." / "_" / "~" / "+" / "/" ) *"="



; ----------------------------------------
;   3GPP TS 29.500
;
;   Version: 18.34.0 (SeptDecember 2023)
;
;   (c) 2023, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
; ----------------------------------------


(… text not shown for clarity …)


;
; Header: 3gpp-Sbi-Oci
;

Sbi-Oci-Header  = "3gpp-Sbi-Oci:" OWS oci-element *( OWS "," OWS oci-element ) OWS

oci-element     = timestamp ";" RWS validityPeriod ";" RWS olcMetric ";" RWS olcScope

timestamp       = "Timestamp:" RWS DQUOTE date-time DQUOTE

validityPeriod  = "Period-of-Validity:" RWS 1*DIGIT "s"

olcMetric       = "Overload-Reduction-Metric:" RWS ( "100" / %x31-39 DIGIT / DIGIT ) "%"

olcScope        = nfProducerScope / nfConsumerScope / scpScope / seppScope

nfProducerScope = (   ( "NF-Instance:" RWS nfinst ) 
                    / ( "NF-Set:" RWS nfset ) 
                    / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                    / ( "NF-Service-Set:" RWS nfserviceset )
                  ) [ ";" RWS sNssaiList ";" RWS dnnList ]

nfConsumerScope = ( "NFC-Instance:" RWS nfinst [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NFC-Set:" RWS nfset [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NFC-Service-Instance:" RWS nfservinst [";" RWS "NF-Inst:" RWS nfinst ] )
                / ( "NFC-Service-Set:" RWS nfserviceset )
                / ( "Callback-Uri:" RWS DQUOTE URI DQUOTE *( RWS "&" RWS DQUOTE URI DQUOTE ) )

servname        = token

scpScope        = "SCP-FQDN:" RWS fqdn

seppScope       = "SEPP-FQDN:" RWS fqdn

fqdn            = token

dnnList         = "DNN:" RWS 1*tchar *( RWS "&" RWS 1*tchar )

sNssaiList      = "S-NSSAI:" RWS snssai *( RWS "&" RWS snssai )

snssai          = 1*tchar


(… text not shown for clarity …)


;
; Header: 3gpp-Sbi-Lci
;

Sbi-Lci-Header    = "3gpp-Sbi-Lci:" OWS lc-element *( OWS "," OWS lc-element ) OWS

lc-element        = timestamp ";" RWS lcMetric ";" RWS lcScope

lcMetric          = "Load-Metric:" RWS ( "100" / %x31-39 DIGIT / DIGIT ) "%"

lcScope           = lcNfProducerScope / scpScope / seppScope

lcNfProducerScope = (   ( "NF-Instance:" RWS nfinst )
                      / ( "NF-Set:" RWS nfset )
                      / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                      / ( "NF-Service-Set:" RWS nfserviceset )
                    ) [ ";" RWS sNssaiList ";" RWS dnnList ";" RWS relativeCapacity ]

relativeCapacity  = "Relative-Capacity:" RWS ( "100" / 1*2DIGIT ) "%"


(… text not shown for clarity …)





;
; Header: 3gpp-Sbi-Selection-Info
;

Sbi-Selection-Info-Header = "3gpp-Sbi-Selection-Info:" OWS
                            selection-info-element *( OWS "," OWS selection-info-element ) OWS

selection-info-element    = ( "reselection=" reselectionvalue *( ";" OWS selection-criteria ) )
                          / ( selection-criteria *( ";" OWS selection-criteria ) )

reselectionvalue          = "true" / "false"

selection-criteria        = selection-action "=" token

selection-action          = "not-select-nfservinst"
                          / "not-select-nfserviceset"
                          / "not-select-nfinst"
                          / "not-select-nfset"


(… text not shown for clarity …)


;
; Header: 3gpp-Sbi-Request-Info
;

Sbi-Request-Info-Header = "3gpp-Sbi-Request-Info:" OWS req-param *( ";" OWS req-param ) OWS

req-param               = req-param-name "=" OWS req-param-value

req-param-name          = "retrans" 
                        / "redirect"
                        / "reason"
                        / "idempotency-key"
                        / "receivedrejectioncause"
                        / “"callback-uri-prefix”"
                        / token

req-param-value         = token


(… text not shown for clarity …)

* * * End of Changes * * * *
